I N T E R P R E T I V E L O G I C

W i l B a d e n
339 Princeton Drive
Costa Mesa CA 92626

A programmer's best friend is his source-code editor. For Forth programmers the relationship can be particularly intimate. A Forth editor will often do very kinky things for the programmer.

The next best friend that a programmer can have is the ability to do conditional interpretation and conditional compilation, both from immediate console entry and from a prepared script. In Unix (tm) parlance this is the Shell.

The Forth-83 Uncontrolled Reference Set has three words, IFTRUE , OTHERWISE , and IFEND , inherited from the Forth-79 Reference Set, whose meaning is specified by:

IFTRUE flag -- interpret only

Begin an

IFTRUE ... OTHERWISE ... IFEND

conditional sequence. These conditional words operate like

IF ... ELSE ... THEN

except that they cannot be nested, and are to be used only during interpretation. In conjunction with the words [and] they may be used within a colon definition to control compilation, although they are not to be compiled.

There is no reason why these could not and should not be nested. If they are STATE-smart they do not have to bear new names, which will make them easier to learn.

One method of implementation is to scan the source input for the relevant key words and perform an appropriate action for each. The string processing routines needed to do this already exist in Forth -- WORD , FIND and EXECUTE . An easy way to restrict the words that are recognized is to use a sealed vocabulary containing just those words and no others.

After IF , ELSE and THEN are defined it is a small step to BEGIN , WHILE , REPEAT and UNTIL . Another small step will lead to DO and LOOP . This extends Forth to become a very powerful job control language.

With a few extensions to the line editing words such as found in Laxen-Perry's F83 Model or Leo Brodie's Starting Forth we get a powerful programmable macro editor.

Some examples of use are found on the load block.

```
CR .( INTERPRETIVE VERSION OF LOGIC STRUCTURES.)  CAPS ON

3 7 THRU  ( IF ELSE THEN  HAVE  BEGIN WHILE REPEAT UNTIL )

STRETCH FORTH  MS-DOS  WWB 851019
```

```
CR .( Do you want interpretive DO ? )
KEY ASCII _ AND   ASCII Y =
if CR .( And do you want interpretive FOR ? )
   KEY ASCII _ AND   ASCII Y =
   if    8 10 THRU       ( DO FOR LOOP I )
   else  9 10 THRU then  ( DO LOOP I )
then

have PRINT not if 11 LOAD then
have CONSULT if 2 2 CONSULT then
```

First we load
IF ELSE THEN BEGIN WHILE REPEAT UNTIL
from blocks 3 through 7. We then ask whether interpretive DO (and LOOP) are wanted. KEY is used to accept your answer, and if it was uppercase or lowercase Y we ask if you also want interpretive FOR . The appropriate blocks are loaded.

IF and ELSE are very useful with conditional compilation. We can compile or execute different things depending on the value of a variable or computed expression. We can compile different systems from just one source program. The word HAVE can be used to test whether we already have a word, i.e., whether it has already been defined.

The code is compatible with the Laxen-Perry F83 Model (and of course my F83X). A simpler, more powerful, and easier to use definition of SEAL is given.

SEAL
Usage: SEAL will change the search order such that only the first vocabulary in the search order will be searched. The other vocabularies in the search order are present and unchanged but do not participate in the search.

UNSEAL
Allow all vocabularies in the search order to be searched.

PASSOVER is used by the system to pass over all words but those in the vocabulary CONDITIONAL . Those words and only those words will be executed.

The word FOR has been introduced for interpretive counted loops. In most applications it is more convenient to express a range as <first>,<last> rather than <last+1>,<first> . When a word is compiled it is trivial to say "1+ SWAP" before "DO" but it becomes a nuisance and a potential source of confusion when interpreting. FOR also checks that there are at least two arguments on the stack. A compilation version of FOR has been given for completeness.

The loop index I presents a problem when the editor words include an I also. We use conditional compilation to check whether the editor does contain I and when it does we define ^ (caret or up-arrow) as a synonym of the editor I for use with

interpretive logic. Conditionally selected system dependent code is used to check the depth of the return stack so that the editor I can still be used outside of interpretive logic.

The use of interpretive logic is simplified when entering commands from the keyboard by not requiring THEN , REPEAT or LOOP at the end of a line. They are required on a source block, or when you want to do something after the interpretion of the control structure, or when compiling. With interpretive logic the Laxen-Perry F83 MANY and TIMES are redundant.

To prevent an infinite loop from happening the interpretive BEGIN and DO will be halted by pressing any key, and terminated if the next key pressed is <return>.

The implementation of +LOOP presents no problem and has been left as an exercise. It just requires more bookkeeping to maintain and update the loop index. I do not feel that this extra overhead would be justified by its frequency of use.

The code and examples are case insensitive. The use of upper and lower case is intended only for clarity

The method of printing this exhibit demonstrates several applications of interpretive logic. The word DOC is used to display the lines of a source block other than the first. In addition, multiple blank lines are suppressed. If DOC has not already been defined then the definition will be made.

This word is then used in interpretive loops to display text. If the listing goes to the terminal rather than a line printer, i.e., the value of variable PRINTING is zero, then you will be asked if you now want to forget the word DOC , which you presumably have just defined, used, and no longer need.

CONCLUSION

Interpretive logic, for conditional execution, conditional compilation, and text editing, extends Forth to be responsive to modern system requirements.

```
have DOC not if
: DOC  ( block# -- ) 1 ?ENOUGH   L/SCR 1
   DO I C/L *   OVER BLOCK +   C/L -TRAILING   ?DUP
      IF CR >TYPE
      ELSE DROP #OUT @
         IF CR THEN
      THEN
   LOOP DROP ;
then
13 16 for i doc loop   1 doc   17 22 for i doc loop   0 fh doc
printing @ not if
   .( FORGET DOC ? )   key ascii _ and ascii Y =
   if FORGET DOC then
then
```

```
    0
 0 ( Interpretive logic structures.                  WWB 19OCT85WWB)
 1
 2              I N T E R P R E T I V E   L O G I C
 3
 4 The logical structure words are extended for interpretation.
 5
 6                       W i l   B a d e n
 7                      339 Princeton Drive
 8                      Costa Mesa CA 92626
 9                          714-546-9894
10
11
12
13
14
15
```

```
    1
 0 ( INTERPRETIVE LOGIC                              WWB/20OCT85WWB)
 1 CR .( INTERPRETIVE VERSION OF LOGIC STRUCTURES.)   CAPS ON
 2 8 VIEW# !   8 VIEWS IF.BLK
 3 3 7 THRU  ( IF ELSE THEN   HAVE   BEGIN WHILE REPEAT UNTIL )
 4 CR .( Do you want interpretive DO ? )
 5 key dup emit   ascii _ and   ascii Y =
 6 if CR .( And do you want interpretive FOR ? )
 7    key dup emit   ascii _ and   ascii Y =
 8    if   8 10 THRU       ( DO FOR LOOP I )
 9    else  9 10 THRU then  ( DO LOOP I )
10 then
11 have PRINT not if 11 LOAD then
12 have CONSULT if 2 2 CONSULT then
13
14
15
```

```
    2
 0 ( DIRECTORY OF LOGICAL INTERPRETATION WORDS.   WWB  20OCT85WWB)
 1 SEAL 4 UNSEAL 4 PASSOVER 4 IF 5 ELSE 5 THEN 5 HAVE 5 HAVE 5
 2 THENIF 5 NUF? 6 BEGIN 7 REPEAT 7 UNTIL 7 WHILE 7 INCL 8 FOR 8
 3 DO 9 LOOP 9 FOR 9 I 10 ^ 10 I 10
 4
 5
 6
 7
 8
 9
10
11
12
13
14
15
```

```
   12
 0 ( SAY "CAPACITY 2/ LOAD" TO PRINT EXHIBIT.        WWB 19OCT85WWB)
have DOC not if
: DOC  ( block# -- ) 1 ?ENOUGH   L/SCR 1
   DO I C/L *   OVER BLOCK +   C/L -TRAILING   ?DUP
      IF CR )TYPE
      ELSE DROP #OUT 2
         IF CR THEN
      THEN
   LOOP DROP ;
then
13 16 for i doc loop   1 doc   17 22 for i doc loop   0 fh doc
printing 2 not if
   .( FORGET DOC ? )   key ascii _ and ascii Y =
   if FORGET DOC then
then
```

```
   13
 0 ( Interpretive logic structures.                  WWB 19OCT85WWB)
                I N T E R P R E T I V E   L O G I C

                       W i l   B a d e n
                      339 Princeton Drive
                      Costa Mesa CA 92626

    A programmer's best friend is his source-code editor. For
Forth programmers the relationship can be particularly
intimate. A Forth editor will often do very kinky things for
the programmer.

    The next best friend that a programmer can have is the
ability to do conditional interpretation and conditional
compilation, both from immediate console entry and from a
prepared script. In Unix (tm) parlance this is the Shell.
```

```
   14
 0 ( Interpretive logic structures.                  WWB 19OCT85WWB)
    The Forth-83 Uncontrolled Reference Set has three words,
IFTRUE , OTHERWISE , and IFEND , inherited from the Forth-79
Reference Set, whose meaning is specified by:

IFTRUE          flag --                          interpret only
   Begin an
      IFTRUE ... OTHERWISE ... IFEND
   conditional sequence. These conditional words operate like
      IF ... ELSE ... THEN
   except that they cannot be nested, and are to be used only
   during interpretation. In conjunction with the words [ and
   ] they may be used within a colon definition to control
   compilation, although they are not to be compiled.
```

3

```
0 ( Interpretive conditional                    WWB/WWB 850725)
1 VOCABULARY CONDITIONAL  CONDITIONAL DEFINITIONS
2 ( PASSOVER will ignore all words but these.)
3 ( You should extend this set if you define other structures.)
4 : else   2DUP = + ;
5 : if     1+ ;
6 : begin  1+ ;
7 : then   1- ;
8 : repeat 1- ;
9 : until  1- ;
10 : ."    [ ASCII " ] LITERAL PARSE 2DROP ;
11 : .(    [COMPILE] ( ;
12 : (     [COMPILE] ( ;
13 : \     [COMPILE] \ ;
14 FORTH DEFINITIONS
15
```

4

```
0 ( Interpretive bypass                         WWB/WWB 850725)
1 : SEAL    ( -- ) 1 IS #VOCS ;                ( Redefined.)
2 : UNSEAL  ( -- ) [ #VOCS ] LITERAL IS #VOCS ;
3
4 : PASSOVER  ( n -- ) 1 ( initial depth)
5   BEGIN BL WORD   DUP COUNT   ?DUP 0=
6      IF 2DROP 2DROP   BLK @ ABORT" Unexpected end." EXIT THEN
7      UPPER   CONTEXT @ >R   SEAL CONDITIONAL   FIND   UNSEAL
8      R> CONTEXT !   IF EXECUTE ?DUP THEN   0=
9   UNTIL DROP ;
10
11
12
13
14
15
```

5

```
0 ( Interpretive if-else-then                   WWB/WWB 850722)
1 : if      ( n -- )
2   STATE @ if [compile] IF exit then   1 ?ENOUGH
3   0= IF 1 PASSOVER THEN ; IMMEDIATE
4 : else    ( -- )
5   STATE @ if [compile] ELSE exit then
6   0 PASSOVER ; IMMEDIATE
7 : then    ( -- )
8   STATE @ if [compile] THEN then  ; IMMEDIATE
9 \ : have  ( -- flag ) BL WORD FIND   SWAP DROP   0= NOT ;
10 : have  ( -- flag ) DEFINED NIP 0<> ;
11
12 have THENIF not if EXIT then
13 : thenif  ( n -- )
14   STATE @ if [compile] THENIF exit then
15   [COMPILE] if ; IMMEDIATE
```

15

(Interpretive logic structures. WWB 19OCT85WWB)

There is no reason why these could not and should not be nested. If they are STATE-smart they do not have to bear new names, which will make them easier to learn.

One method of implementation is to scan the source input for the relevant key words and perform an appropriate action for each. The string processing routines needed to do this already exist in Forth -- WORD , FIND and EXECUTE . An easy way to restrict the words that are recognized is to use a sealed vocabulary containing just those words and no others.

After IF , ELSE and THEN are defined it is a small step to BEGIN , WHILE , REPEAT and UNTIL . Another small step will lead to DO and LOOP . This extends Forth to become a very powerful job control language.

16

(Interpretive logic structures. WWB 19OCT85WWB)

With a few extensions to the line editing words such as found in Laxen-Perry's F83 Model or Leo Brodie's Starting Forth we get a powerful programmable macro editor.

Some examples of use are found on the load block.

17

(Interpretive logic structures. WWB 19OCT85WWB)

First we load

IF ELSE THEN BEGIN WHILE REPEAT UNTIL

from blocks 3 through 7. We then ask whether interpretive DO (and LOOP) are wanted. KEY is used to accept your answer, and if it was uppercase or lowercase Y we ask if you also want interpretive FOR . The appropriate blocks are loaded.

IF and ELSE are very useful with conditional compilation. We can compile or execute different things depending on the value of a variable or computed expression. We can compile different systems from just one source program. The word HAVE can be used to test whether we already have a word, i.e., whether it has already been defined.

```
     6
 0 ( NUF?                                        MJT/19OCT85WWB)
 1 have NUF? not if
 2 : NUF?  ( -- flag )
 3    KEY? DUP IF KEY 2DROP   KEY 13 ( return) = THEN ;
 4 then
 5
 6 ( NUF? halts after one key-press, and then returns TRUE if the )
 7 ( next key-press is <return>.  Suggested by Martin Tracy.      )
 8
 9 have FH not if
10 : FH  ( u -- block#) BLK @ ?DUP 0= IF SCR @ THEN + ;
11 then
12
13 ( Relative block number.  Suggested by Leo Brodie.            )
14
15
```

```
     7
 0 ( Interpretive begin-while-repeat-until       WWB/WWB 850722)
 1 : begin   ( -- )
 2    STATE @ if [compile] BEGIN exit then   >IN @ >R
 3    BEGIN R@ >IN !   INTERPRET   NUF? UNTIL R> DROP ; IMMEDIATE
 4 : repeat  ( -- )
 5    STATE @ if [compile] REPEAT exit then
 6    R> DROP ; IMMEDIATE
 7 : until   ( n -- )
 8    STATE @ if [compile] UNTIL exit then   1 ?ENOUGH
 9    IF R> R> 2DROP THEN R> DROP ; IMMEDIATE
10 : while   ( n -- )
11    STATE @ if [compile] WHILE exit then   1 ?ENOUGH
12    0=
13    IF R> R> 2DROP   R> DROP   0 PASSOVER THEN ; IMMEDIATE
14
15
```

```
     8
 0 ( Compiled for-loop                           WWB/09AUG85WWB)
 1 : INCL  ( n1,n2 -- n2+1,n1 ) 2 ?ENOUGH   OVER MAX   1+ SWAP ;
 2 : FOR  ( first,last -- )
 3    COMPILE INCL   [COMPILE] DO ; IMMEDIATE
 4
 5 ( In most applications it is more convenient to express a range)
 6 ( as <first>,<last> rather than <last+1>,<first> -- especially )
 7 ( when interpreting.  This definition of FOR is in anticipation)
 8 ( of an interpretive version;  otherwise it is not worth doing.)
 9
10
11
12
13
14
15
```

```
    18
 0 ( Interpretive logic structures.              WWB 19OCT85WWB)
```

The code is compatible with the Laxen-Perry F83 Model (and of course my F83X). A simpler, more powerful, and easier to use definition of SEAL is given.

SEAL
Usage: SEAL will change the search order such that only the first vocabulary in the search order will be searched. The other vocabularies in the search order are present and unchanged but do not participate in the search.

UNSEAL
Allow all vocabularies in the search order to be searched.

PASSOVER is used by the system to pass over all words but those in the vocabulary CONDITIONAL . Those words and only

```
    19
 0 ( Interpretive logic structures.              WWB 19OCT85WWB)
```

those words will be executed.

The word FOR has been introduced for interpretive counted loops. In most applications it is more convenient to express a range as <first>,<last> rather than <last+1>,<first> . When a word is compiled it is trivial to say "1+ SWAP" before "DO" but it becomes a nuisance and a potential source of confusion when interpreting. FOR also checks that there are at least two arguments on the stack. A compilation version of FOR has been given for completeness.

The loop index I presents a problem when the editor words include an I also. We use conditional compilation to check whether the editor does contain I and when it does we define ^ (caret or up-arrow) as a synonym of the editor I for use with

```
    20
 0 ( Interpretive logic structures.              WWB 19OCT85WWB)
```

interpretive logic. Conditionally selected system dependent code is used to check the depth of the return stack so that the editor I can still be used outside of interpretive logic.

The use of interpretive logic is simplified when entering commands from the keyboard by not requiring THEN , REPEAT or LOOP at the end of a line. They are required on a source block, or when you want to do something after the interpretion of the control structure, or when compiling. With interpretive logic the Laxen-Perry F83 MANY and TIMES are redundant.

To prevent an infinite loop from happening the interpretive BEGIN and DO will be halted by pressing any key, and terminated if the next key pressed is <return>.

```
     9
 0 ( Interpretive do & for                          WWB/09AUG85WWB)
 1 : do  ( limit,start -- )
 2    STATE @ if [compile] DO exit then   2 ?ENOUGH
 3    >IN @ -ROT
 4    DO DUP >IN ! >R   INTERPRET   R> NUF? ?LEAVE LOOP
 5    DROP ; IMMEDIATE
 6 : loop  ( -- )
 7    STATE @ if [compile] LOOP exit then
 8    R> DROP ; IMMEDIATE
 9
10 have FOR not if EXIT then
11 : for  ( first,last -- )
12    STATE @ if [compile] FOR exit then
13    >IN @ -ROT
14    FOR DUP >IN ! >R   INTERPRET   R> NUF? ?LEAVE LOOP
15    DROP ; IMMEDIATE
```

```
     10
 0 ( Interpretive loop-index                        WWB/30JUL85WWB)
 1 : i   ( -- loop-index )
 2    STATE @ if [ forth ] COMPILE I exit then
 3    R> R> R> [ forth ] I SWAP >R SWAP >R SWAP >R ; IMMEDIATE
 4
 5 ' I EDITOR ' I = NOT if DEFINITIONS
 6    : ^   ( -- ) [ editor ] I ; ( In interpreted DO-loops.)
 7    : I   ( -- loop-index )
 8       STATE @ if [ forth ] [compile] I exit then
 9       ( *** RP0 and RP@ are implementation dependent. *** )
10       [ have RP0 if ] RP0 @ [ else ] R0 @ [ then ]
11       RP@ 12 ( 6 CELLS) + U< IF [ editor ] I EXIT THEN
12       R> [ forth ] [compile] i SWAP >R ; IMMEDIATE
13 then FORTH DEFINITIONS
14
15
```

```
     11
 0 ( SINGLE LINESPACE                               WWB 20OCT85WWB)
 1 5 CONSTANT PRINT-OFFSET
 2 : LINESPACE ( -- ) PRINTING @
 3    IF 13 ( CR) (PRINT)
 4       [ .( Do you want LF after CR ? )
 5       Key dup emit   ascii _ and ascii Y =
 6       if ] 10 ( LF) (PRINT) [ then ]
 7       PRINT-OFFSET SPACES
 8    THEN
 9    13 ( CR) CONSOLE   10 ( LF) CONSOLE
10    #OUT OFF  1 #LINE +! ;
11
12
13
14
15
```

     21

(Interpretive logic structures. WWB 19OCT85WWB)

The implementation of +LOOP presents no problem and has been left as an exercise. It just requires more bookkeeping to maintain and update the loop index. I do not feel that this extra overhead would be justified by its frequency of use.

The code and examples are case insensitive. The use of upper and lower case is intended only for clarity

The method of printing this exhibit demonstrates several applications of interpretive logic. The word DOC is used to display the lines of a source block other than the first. In addition, multiple blank lines are suppressed. If this word or another of the same name has not already been defined then the definition will be made.

     22

(Interpretive logic structures. WWB 20OCT85WWB)

This word is then used in interpretive loops to display text. If the listing goes to the terminal rather than a line printer, i.e., the value of the variable PRINTING is zero, then you will be asked if you now want to forget the word DOC which you presumably have just defined, used, and no longer need.

CONCLUSION

Interpretive logic, for conditional execution, conditional compilation, and text editing, extend Forth to be responsive with modern system requirements.

     23

(SINGLE LINESPACE WWB 20OCT85WWB)

LINESPACE (--)

Configure vertical spacing for your printer with interactive conditional compilation.